PharmaSUG 2021 - Paper AP-088

Using SAS for Forest Plots in AMNOG Meta-Analysis Reporting

Jian-An Lu, Independent SAS Consultant

Jinwei Yuan, Vifor Pharma Inc

ABSTRACT

Making forest plots is an indispensable part of presenting study results in a Benefit Dossier for the AMNOG assessment. It allows one to intuitively and quickly display overall efficacy and safety outcomes in a comprehensive manner. This is particularly important given the scope of the assessment, as it involves a large volume of statistics with multiple endpoints, multiple studies and multiple subgroups at both individual and meta-analysis levels. This paper demonstrates, step by step, how to utilize PROC SGPLOT combined with the attribute map and annotation capabilities provided by SAS® to create table-like graphic displays that present all required statistical information in one place as per the IQWiG guidelines. We hope this approach can serve as a primer that can aid in the creation of the sophisticated forest plots required in the AMNOG submission.

KEY WORDS

AMNOG, Forest Plots, Meta-analysis, Relative Risk, Risk Difference, Odds Ratio, Heterogeneity

INTRODUCTION

AMNOG (Arzneimittelmarktneuordnungsgesetz in German) is a law passed by the Federal Parliament (Bundestag) of Germany in 2010, which applies to all pharmaceutical products with active ingredients that are launched beginning January 1, 2011. The law describes the process to determine the price at which an approved new product will be reimbursed by the statutory health insurance system. At the product launch, a manufacturer must submit a Benefit Dossier intended to show that the new product has additional benefit against a comparator (zweckmäßige Vergleichstherapie, zVT) determined by the Federal Joint Committee (Gemeinsamer Bundesausschuss, G-BA). The G-BA retains the Institute for Quality and Efficiency in Healthcare (Institut für Qualität und Wirtschaftlichkeit im Gesundheitswesen, or IQWiG) to conduct benefit assessment.

Different from the assessments underlining regulatory decisions, a Benefit Dossier for the G-BA focuses on the assessment of an additional benefit for the new product compared to an appropriate comparator therapy (ACT) (Leverkus F et al., 2016).

GRAPH REQUEST

In a hypothetical AMNOG submission case, two identical Phase 3 pivotal trials (N=100 in the active and placebo arms, respectively in each study) are determined for the benefit assessment in the Benefit Dossier. A typical efficacy table for a binary endpoint is as follows:

Table 1. Number (n, %) of Subjects Who Attain Defined Efficacy Response (ITT Population)

	Study 0001		Study 0002		Meta-Analysis	
	Active	Placebo	Active	Placebo	Active	Placebo
n/N (%)	17/100 (17.0)	8/100 (8.0)	22/100 (22.0)	5/100 (5.0)	39/200 (19.5)	13/200 (6.5)
RR [95% CI), p-value	2.13 [0.96, 4.70], 0.0626		4.40 [1.73, 11.16], 0.0018		3.00 [1.65, 5.45], 0.0003	
OR [95% CI), p-value	2.36 [0.97, 5.74], 0.0543		5.36 [1.94, 14.80], 0.0004		3.48 [1.80, 6.76], 0.0001	
RD [95% CI), p-value	0.09 [-0.00, 0.18], 0.0521		0.17 [0.08, 0.26], 0.0003		0.13 [0.07, 0.19], <0.0001	
Heterogeneity (RR)						
p-value					0.2434	
I^2					26.5%	
Heterogeneity (OR)						
p-value					0.2295	

Abbreviations: CI: Confidence Interval; ITT: Intention-To-Treat; OR: Odds Ratio; RD: Risk Difference; RR: Relative Risk.

Statistical methods: The meta-analysis was conducted by pooled individual patient data without adjustment. If there is no responder in one treatment group, 0.5 is added for calculation. The 95% CI is calculated by the Wald method. The heterogeneity p-value (RR) and I^2 are calculated from lnRR using the fixed-effect model (M Borenstein et al., 2009). The heterogeneity p-value (OR) is based on the Breslow-Day test for homogeneity of the odds ratios between studies. Estimates of Relative Risk>1, Odds Ratio>1 and Risk Difference>0 are in favor of the active treatment.

A figure of forest plots is requested to display all intermediate and inference statistics for Relative Risk as follows:

Figure 1. Forest Plot for Subjects Who Attain Defined Efficacy Response ITT Population



Before programming the forest plot, prepare all required statistics in a dataset shown in Table 2, where the variable descriptions are listed in Table 3.

Table 2. Graph dataset

Id	StudyId	Order	RR	lowerCL	upperCL	pValue	zValue	TE	seTE	Weight	i2	HeterpRR
2	STUDY0001	1	2.13	0.96	4.7	0.0626	1.8623	0.75	0.4048	57.9		
2	STUDY0002	2	4.4	1.73	11.16	0.0018	3.1203	1.48	0.4748	42.1		
1	STUDYMETA	3	3	1.65	5.45	0.0003	3.6109	1.1	0.3042	100	26.5	0.2434

Table 3. Dataset variables and their descriptions

Variable Name	Variable Description
Id	Identifier used by the attribute map
StudyId	Study Identifier
Order	Observation order internally used
RR	Relative Risk
lowerCL	Lower limit of 95% confidence interval for RR
upperCL	Upper limit of 95% confidence interval for RR
pValue	p-values for the testing of RR
zValue	Z statistic for p-value
TE	Treatment effect (lnRR)
seTE	Standard error for TE
Weight	Study Weight
I^2	Information for heterogeneity
HeterpRR	p-value for heterogeneity based on RR

GRAPH PROGRAMMING

Suppose a request is received to create Figure 1. One can perform requirement analysis to identify specific graphing features in the figure and then map them into the SAS graphing functionalities (Table 4). The mapping may vary with a change of the SAS version to use and/or personal preferences.

Table 4. Mapping graphing features to SAS graphing functionalities

	Graphing Feature	SAS Graphing Functionality
1	Statistics on the right of the figure, aligned with y-axis	YAXISTABLE statement in SAS 9.4 and later versions
2	Statistics on the left of the figure which can be lined up in 3 columns, aligned with y-axis	YAXISTABLE statement in SAS 9.4 and later versions
3	Different font used for the meta-analysis statistics from the font used for individual study statistics	DATTRMAP option of PROC SGPLOT
4	Statistics on the left of the figure which CAN NOT be nicely lined up in columns, and different fonts are used in a single line	PROC SGPLOT annotation in SAS 9.3 and later versions
5	Label (Risk Ratio) on top of the vertical line with x-axis value = 1	X2AXIS in PROC SGPLOT
6	Line to represent the 95%-CI of the relative risk.	HIGHLOW statement in PROC SGPLOT
7	Dot and square centered by the dot to represent the relative risk and the weight	SCATTER statement in PROC SGPLOT
8	Diamond to represent the meta-analysis relative risk and its 95%-CI	PROC SGPLOT annotation in SAS 9.3 and later versions
9	X-axis scale in natural log scale	XAXIS statement in PROC SGPLOT
10	Text underneath x-axis	PROC SGPLOT annotation in SAS 9.3 and later versions
11	Vertical reference lines at x = 1 and x = meta-analysis relative risk, respectively with different line styles	REFLINE statement in PROC SGPLOT

After the determination of above SAS functionalities, a SAS program can be developed in the following 5 major steps.

Step 1. Manipulate the dataset in Table 2.

1. Convert values of the numeric variables RelativeRisk, lowerCL, upperCL, TE, weight, and seTE into character values with preferred precisions and save them in new character variables c1 – c6 in order.
2. Create a character variable ci using c2 and c3 to represent the 95%-Cis
3. Adjust and align above character values for cosmetic reasons.
4. Move relative risk values of individual studies to new variables rr1 and rr2, respectively, because they need to be drawn with different scatter statements due to the different square sizes.
5. Save values of relative risk, lower and upper confidence limits of the meta-analysis into macro variables to be used in annotation dataset to draw the diamond.
6. Create a numeric variable GRP for y-axis values to be used in creating the figure.
7. Create a character variable STUDY with texts to be shown on the figure.

Below are the variables in the derived dataset to be used to draw the figure.

	Id	Order	pValue	zValue	i2	HeterpRR	c1	c4	c5	c6	ci	study	grp	rr1	rr2	rr3	lcl3	ucl3
1	2	1	0.0626	1.8623			2.13	0.75	57.9%	0.4048	[0.96, 4.70]	STUDY0001	4	2.13				
2	2	2	0.0018	3.1203			4.40	1.48	42.1%	0.4748	[1.73, 11.16]	STUDY0002	3		4.4			
3	1	3	0.0003	3.6109	26.5	0.2434	3.00	1.10	100.0%	0.3042	[1.65, 5.45]	Fixed effect model	1			3	1.65	5.45

Step 2. Create dataset ATTRMAP to use bold font for meta-analysis statistics and normal font for individual study statistics.

	textweight	id	value	textcolor	textsize
1	bold	text	1	Black	10
2	normal	text	2	Black	10

Step 3. Create dataset ANNO containing the annotation information.

	function	anchor	drawspace	display	fillcolor	heightunit	widthunit	layer	textstyle	textfont	x1space	y1space	label	x1	y1	filltransparency	textsize	linethickness	width
1	polygon		datavalue	all	gray	Data	Data	back		Arial				1.6525276112	1	0.5	8	2	
2	polycont		datavalue		gray	Data	Data	back		Arial				3	0.8	0	8	2	
3	polycont		datavalue		gray	Data	Data	back		Arial				5.4462024954	1	0	8	2	
4	polycont		datavalue		gray	Data	Data	back		Arial				3	1.2	0	8	2	
5	text	left						front		Arial	graphpercent	datavalue	Heterogeneity :	0.2	0.1	0	8	2	60
6	textcont		datavalue			Data	Data		italic	Arial			("ESC")(unicode '00A0'x) i			0	8	2	
7	textcont		datavalue			Data	Data			Arial			("ESC")(sup '2')			0	8	2	
8	textcont		datavalue			Data	Data			Arial			("ESC")(unicode '00A0'x)= 26.5%,			0	8	2	
9	textcont		datavalue			Data	Data		italic	Arial			("ESC")(unicode '00A0'x)p			0	8	2	
10	textcont		datavalue			Data	Data			Arial			("ESC")(unicode '00A0'x) = 0.2434			0	8	2	
11	text	left						front		Arial	graphpercent	datavalue	Overall FEM:	0.2	-0.7	0	8	2	60
12	textcont		datavalue			Data	Data		italic	Arial			("ESC")(unicode '00A0'x)z			0	8	2	
13	textcont		datavalue			Data	Data			Arial			("ESC")(unicode '00A0'x)= 3.61 (			0	8	2	
14	textcont		datavalue			Data	Data		italic	Arial			p =			0	8	2	
15	textcont		datavalue			Data	Data			Arial			("ESC")(unicode '00A0'x)0.0003)			0	8	2	
16	text	center								Arial	graphpercent	datavalue	Favours Placebo Arm Favours Active Arm	51	-2.5	0	8	2	60

Step 4. Calculate marker sizes used in SCATTER statement for creating squares whose areas represent weights. The actual side lengths would need to be adjusted appropriately by different ways.

Step 5. Use PROC SGPLOT to create the figure.

```
ods graphics / reset width=5.5in height=1.5in border=off;
proc sgplot data=m1 noautolegend nocycleattrs pad=(bottom=75) dattrmap=attrmap nowall noborder
                    sganno=anno;
  refline 1 / axis=x lineattrs=(pattern=solid) transparency=0.1;
  refline rr3 / axis=x lineattrs=(pattern=shortdash) transparency=0.5;
  highlow y=grp low=lowercl high=uppercl / lowcap=none highcap=none lineattrs=(color=black);
  scatter y=grp x=lowercl / markerattrs=(size=0) x2axis;
  scatter y=grp x=rr1  / markerattrs=(size=&ht1 symbol=squarefilled color=gray) transparency=0.5
                    markeroutlineattrs=(thickness=0);
  scatter y=grp x=rr2  / markerattrs=(size=&ht2 symbol=squarefilled color=gray) transparency=0.5
                    markeroutlineattrs=(thickness=0);
```

```
  scatter y=grp x=relativerisk / markerattrs=(size=4 symbol=circlefilled);
  yaxistable c1 ci c5 / location=outside position=right labelattrs=(size=11 weight=bold family=Arial)
                        valueattrs=(family=Arial size=10) textgroup=id textgroupid=text pad=(right=10px)
                        valuejustify=right labeljustify=center;
  yaxistable study / location=outside position=left labelattrs=(size=11 weight=bold family=Arial)
                        valueattrs=(family=Arial size=10) textgroup=id textgroupid=text pad=(right=60px)
                        valuejustify=left labeljustify=left;
  yaxistable c4 c6 / location=outside position=left labelattrs=(size=11 weight=bold family=Arial)
                        valueattrs=(family=Arial size=10) textgroup=id textgroupid=text pad=(right=10px)
                        valuejustify=right labeljustify=center;
  xaxis type=log logbase=e min=0.01 max=200 minor values=(&val2.) valueattrs=(size=8)
                        labelattrs=(size=8) offsetmin=0 offsetmax=0 display=(nolabel)
                        labelattrs=GraphUnicodeText;
  yaxis display=none offsetmin=0.1 offsetmax=0.05 values=(0 to 5);

  x2axis label='Risk Ratio' display=(noline noticks novalues)
                       labelattrs=(size=11 family=Arial weight=bold);* valueattrs=(size=5);
run;
```

CONCLUSION

In the above paper, we have described detailed SAS procedures that generate a forest plot graph commonly requested in the AMNOG submission. A typical AMNOG forest plot figure consists of both intermediate and inference statistics in a standardized comparison (e.g. relative risk) among treatment arms, studies and subgroups as well as in the overall (meta-analysis level). Of note, the example here is only an illustration for an “ideal” data scenario and display. Several special issues which commonly occur can be quite challenging and are not discussed in the paper. Those issues include, but are not limited to, the following:

1. Multiple forest plots are placed in one file but with a large difference in the effect magnitude (e.g. different adverse event)
2. Forest plots in some subgroups with scarce or missing data
3. PDF converting
4. AXISTABLE statement currently does not seem to support ODS escape character

We encourage readers to customize the above SAS code by adding more specific components based on the basic framework and tips provided in this paper as well as some useful references.

APPENDIX

```
*************************************************************************
**  Forest Plots for AMNOG Meta-analysis Reporting
**  SAS 9.4 or later
**  Name   :  Jian-An Lu, Jinwei Yuan
**  Date   :  22MAR2021
*************************************************************************;


*************************************************************************
**  Create a dataset containing required statistics
*************************************************************************;
data forest;
   input Id StudyId $3-11 Order RR lowerCL upperCL pValue zValue TE seTE
         Weight i2 HeterpRR;
   datalines;
2 STUDY0001 1 2.13 0.96  4.70 0.0626 1.8623 0.75 0.4048  57.9    .      .
2 STUDY0002 2 4.40 1.73 11.16 0.0018 3.1203 1.48 0.4748  42.1    .      .
1 STUDYMETA 3 3.00 1.65  5.45 0.0003 3.6109 1.10 0.3042 100   26.5 0.2434
;
run;

data forest1(drop=i);
   set forest;
   length c1 - c6 $20;
   array cols (4) rr lowercl uppercl te;
   array ccs (4) $ c1 - c4;
   array lens (4) l1-l4;
   do i=1 to 4;
      ccs[i]=strip(put(cols[i],10.2));
      lens[i]=length(ccs[i]);
   end;
   if not missing(weight) then do;
      c5=strip(put(weight,10.1)); l5=length(c5);
   end;
   if not missing(sete) then do;
      c6=strip(put(sete,12.4)); l6=length(c6);
   end;
run;

proc sql noprint;
   create table forest2 as
   select *, max(l1) as ml1, max(l2) as ml2, max(l3) as ml3, max(l4) as ml4,
          max(l5) as ml5, max(l6) as ml6 from forest1
   group by order;
quit;

data final(drop=l1-l6 i chklen ml:);
   set forest2;
   array ccs (6) c1-c6;
   array lens (6) l1-l6;
   array mlens (6) ml1-ml6;
   do i=1 to 6;
      if mlens[i]-lens[i]>0 then
         ccs[i]=cats(repeat('A0'x,mlens[i]-lens[i]),ccs[i]);
   end;
```

```
   length ci study $30;
   ci=cats(repeat('A0'x,4),'[',c2,',','A0'x,c3,']','A0'x,'A0'x);
   chklen=length(c2);
   c1=cats(repeat('A0'x,2),c1);
   c4=cats(repeat('A0'x,2),c4);
   c5=cats(repeat('A0'x,2),c5,'%');
   c6=cats(repeat('A0'x,2),c6);
   study=cats(study,repeat('A0'x,2));
   if order=1 then do; grp=4; id=2; study='STUDY0001'; rr1=rr; end;
   if order=2 then do; grp=3; id=2; study='STUDY0002'; rr2=rr; end;
   if order=3 then do;
      grp=1; id=1; study='Fixed effect model'; rr3=rr; lcl3=lowercl;
      ucl3=uppercl; rr=.; lowercl=.; uppercl=.;
   end;
   label c1='  RR' ci='   95%-CI' c5='Weight' c4='  TE' c6='  seTE'
         study='Study';
run;

proc sort data=final;
   by descending grp;
run;

** Create an attribute map dataset for text attributes in yaxistables **;
data attrmap;
   length textweight $10;
   id='text'; value='1'; textcolor='Black'; textsize=10; textweight='bold';
   output;
   id='text'; value='2'; textcolor='Black'; textsize=10; textweight='normal';
   output;
run;


*************************************************************************
**  Create macro variables whose values are to be used in the forest plot
*************************************************************************;

** to adjust horizontal position of the label under x-axis ticks **;
%let x1=52;

*************************************************************************
** log(x) =    -6    -5   -4   -3   -2   -1   0   1   2   3   4   5   6
**      x = 0.003 0.007 0.02 0.05  0.1  0.4   1   3   7  20  55 150 400
*************************************************************************;

/* x-axis tick values */
%let val1=0.007 0.02 0.05 0.1 0.4 1 3 7 20 55 150;

**  title 1  **;
%let t1=Figure 1. Forest Plot for Subjects Who Attain Defined Efficacy
Response;

**  macro variables to be used in annotation dataset or proc sgplot  **;
data _null_;
   set final;
   where studyid='STUDYMETA';
   call symput('rr',rr3);
   call symput('lcl',lcl3);
```

```
   call symput('ucl',ucl3);
   call symput('zval',strip(put(zval,8.2)));
   if pvalue<0.0001 then call symput('opval','<0.0001');
   else call symput('opval',cat('= ',strip(put(pval,8.4))));
   call symput('i2',strip(put(i2,6.1)));
   if heterp_rr<0.0001 then call symput('hpval','<0.0001');
   else call symput('hpval',strip(put(heterp_rr,10.4)));
   length pvalc zvalc $6;
   pvalc=strip(put(pvalue,10.4));
   zvalc=strip(put(zvalue,10.2));
   call symput('opval',pvalc);
   call symput('zval',zvalc);
run;

**   calculate square side sizes   **;
data ht;
   set final(keep=studyid order weight);
   where studyid in ('STUDY0001','STUDY0002');
run;

proc sort data=ht;
   by order;
run;

proc transpose data=ht out=ht_t;
   var weight;
run;

data _null_;
   set ht_t;
   c1=sqrt(col1/100); c2=sqrt(col2/100);
   unit=48/(c1+c2);
   h1=round(c1*unit); h2=round(c2*unit);
   call symput('ht1',h1);
   call symput('ht2',h2);
run;

*************************************************************************
**  Create annotation dataset
*************************************************************************;

data anno;
   length function anchor drawspace display fillcolor heightunit widthunit
          layer textstyle textfont $9 x1space y1space $14 label $60;
   function='polygon'; drawspace='datavalue'; display='all';
      fillcolor='gray'; heightunit='Data'; widthunit='Data'; layer='back';
      x1=&lcl; y1=1; filltransparency=0.5; textfont='Arial'; textsize=8;
      linethickness=2; output;
   function='polycont'; display=''; x1=&rr; y1=0.8; filltransparency=0;
      output;
   x1=&ucl; y1=1; output;
   x1=&rr; y1=1.2; output;
   function='text'; anchor='left'; drawspace=''; heightunit=''; widthunit='';
      fillcolor=''; layer='front'; x1space='graphpercent';
      y1space='datavalue'; x1=0.2; y1=0.1; label='Heterogeneity :'; width=60;
      output;
   function='textcont'; anchor=''; drawspace='datavalue'; heightunit='Data';
```

```
      widthunit='Data'; layer=''; x1space=''; y1space=''; textstyle='italic';
      x1=.; y1=.; label="(*ESC*){unicode '00A0'x} i"; width=.; output;
   textstyle=''; label="(*ESC*){sup '2'}"; output;
   label="(*ESC*){unicode '00A0'x}= &i2.%,"; output;
   textstyle='italic'; label="(*ESC*){unicode '00A0'x}p"; output;
   textstyle=''; label="(*ESC*){unicode '00A0'x} = &hpval"; output;
   function='text'; drawspace=''; anchor='left'; heightunit=''; widthunit='';
      layer='front'; x1space='graphpercent'; y1space='datavalue';
      x1=0.2; y1=-0.7; label='Overall FEM:'; width=60; output;
   function='textcont'; drawspace='datavalue'; anchor=''; heightunit='Data';
      widthunit='Data'; layer=''; x1space=''; y1space=''; textstyle='italic';
      x1=.; y1=.; label="(*ESC*){unicode '00A0'x}z"; width=.; output;
   textstyle=''; label="(*ESC*){unicode '00A0'x}= &zval ("; output;
   textstyle='italic'; label='p = '; output;
   textstyle=''; label="(*ESC*){unicode '00A0'x}&opval.)"; output;
   function='text'; anchor='center'; drawspace=''; heightunit='';
      widthunit=''; x1space='graphpercent'; y1space='datavalue';
      x1=&x1; y1=-2.5; label='Favours Placebo Arm     Favours Active Arm';
      width=60; output;
run;


options missing=' ' orientation=portrait nodate nonumber;
ods _all_ close;
ods pdf file="<Place your path here>\ForestPlot.pdf" nogtitle startpage=no;
ods graphics / reset width=5.5in height=1.5in border=off;
ods escapechar='^';

title1 j=center height=10pt font=Arial bold "&t1.";
title2 j=center height=10pt font=Arial bold "ITT Population";


proc sgplot data=final noautolegend nocycleattrs pad=(bottom=75)
                dattrmap=attrmap nowall noborder sganno=anno;
   refline 1 / axis=x lineattrs=(pattern=solid) transparency=0.1;
   refline rr3 / axis=x lineattrs=(pattern=shortdash) transparency=0.5;
   highlow y=grp low=lowercl high=uppercl / lowcap=none highcap=none
           lineattrs=(color=black);
   scatter y=grp x=lowercl / markerattrs=(size=0) x2axis;
   scatter y=grp x=rr1 / markerattrs=(size=&ht1 symbol=squarefilled
           color=gray) transparency=0.5 markeroutlineattrs=(thickness=0);
   scatter y=grp x=rr2 / markerattrs=(size=&ht2 symbol=squarefilled
           color=gray) transparency=0.5 markeroutlineattrs=(thickness=0);
   scatter y=grp x=rr  / markerattrs=(size=4 symbol=circlefilled);
   yaxistable c1 ci c5 / location=outside position=right labelattrs=(size=11
           weight=bold family=Arial) valueattrs=(family=Arial size=10)
           textgroup=id textgroupid=text pad=(right=10px) valuejustify=right
           labeljustify=center;
   yaxistable study / location=outside position=left labelattrs=(size=11
           weight=bold family=Arial) valueattrs=(family=Arial size=10)
           textgroup=id textgroupid=text pad=(right=60px) valuejustify=left
           labeljustify=left;
   yaxistable c4 c6 / location=outside position=left labelattrs=(size=11
           weight=bold family=Arial) valueattrs=(family=Arial size=10)
           textgroup=id textgroupid=text pad=(right=10px) valuejustify=right
           labeljustify=center;
   xaxis  type=log logbase=e min=0.01 max=200 minor values=(&val1.)
```

```
            valueattrs=(size=8) labelattrs=(size=8)
            offsetmin=0 offsetmax=0 display=(nolabel)
            labelattrs=GraphUnicodeText;
    yaxis  display=none offsetmin=0.1 offsetmax=0.05 values=(0 to 5);

    x2axis label='Risk Ratio' display=(noline noticks novalues)
            labelattrs=(size=11 family=Arial weight=bold);
run;

ods graphics off;
ods _all_ close;
ods listing;
```

REFERENCES

Base SAS® 9.4 Procedures Guide: Statistical Procedures, Sixth Edition

SAS® 9.4 ODS Graphics: Procedures Guide, Sixth Edition

Kuhfeld, W. F. (2016). Basic ODS Graphics Examples. SAS Institute Inc.

Kuhfeld, W. F. (2015). Advanced ODS Graphics Examples. SAS Institute Inc.

Leverkus F, Chuang-Stein C. (2016) Implementation of AMNOG: An industry perspective. Biometrical Journal 58: 76-88.

Borenstein, M., Hedges, L.V., Higgins, J.P.T., Rothstein H.R. (2009). Introduction to Meta-Analysis

ACKNOWLEDGMENTS

We would like to thank Vifor Pharma for sponsoring this work and supporting conference participation. We would like to sincerely thank Dr. Susan Arthur for her guidance and suggestions for this paper.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Name: Jinwei Yuan

Enterprise: Vifor Pharma, Inc

Address: 100 Cardinal Way

City, State ZIP: Redwood City, CA, 94063

Work Phone: 650-421-9675

E-mail: Jinwei.yuan@viforpharma.com

Web: www.viforpharma.com/us